

AMPL: Esempi e Comandi Avanzati

F. Rinaldi

Dipartimento di Matematica
Università di Padova

Corso di Laurea Matematica

Outline

AMPL: Esempi e Comandi Avanzati

Comandi Avanzati

Script per Operazioni Complesse

Ciclo For

```
for {e in INSIEME}{...}
```

Ciclo Repeat While (termina se espressione logica falsa)

```
repeat while espressionelogica {...};  
repeat {...}while espressionelogica;
```

Ciclo Repeat Until (termina se espressione logica vera)

```
repeat until espressionelogica {...};  
repeat {...}until espressionelogica;
```

If-Then-Else

```
if espressionelogica then {...}  
    else {...}
```

Script per Operazioni Complesse II

- ▶ **Assegnazione:** let (es. let c:=3;)
- ▶ **Bloccare \sbloccare variabili:** fix \unfix
- ▶ **Ignorare \ripristinare particolari vincoli e funzioni obiettivo:** drop \restore
- ▶ **Stampare a video messaggi:** printf (stessa sintassi C)
script: printf " MESSAGGIO\n"; a video: MESSAGGIO

Esempio 1

Produzione di Parquet alla LegnoPadova SpA

LegnoPadova SpA é un'azienda specializzata nella produzione di parquet. I suoi stabilimenti producono vari tipi di parquet che vengono poi venduti sul mercato generando un determinato profitto. Ciascun settore della produzione puá lavorare un numero limitato di ore secondo la Tabella 1. Nella Tabella 2 si riporta il profitto unitario e la domanda massima per tipologia di parquet:

Tabella 1

Settore	Ore
Falegnameria	2000
Verniciatura	1500
Assemblaggio	1700
Verifica	300
Imballaggio	500

Tabella 2

Tipo	Profitto	Domanda
Supereconomy	10	100
Economy	18	80
Biocity	20	60
Vintage	25	40
Gaia	27	40
Family	28	20
Extralux	35	20

Riportiamo infine le ore di lavorazione necessarie per ciascuna tipologia di parquet in Tabella 3. Determinare il piano produttivo che massimizza il profitto totale.

Tabella 3

	Supereconomy	Economy	Biocity	Vintage	Gaia	Family	Extralux
Falegnameria	7	5	9	10	10	12	15
Verniciatura	2	2	2	3	3	3	3
Assemblaggio	2	2	4	7	9	15	18
Verifica	1	1	1	2	1	2	2
Imballaggio	1	1	1	1	2	1	0

Modello AMPL

rip_ris.mod

```
# INSIEMI DEL PROBLEMA
# insieme dei prodotti
set prodotti;
# insieme delle risorse disponibili
set risorse;

# PARAMETRI DEL PROBLEMA
# profitto per unità venduta
param c{prodotti};
# domanda massima
param d{prodotti};
# quantità di risorsa per unità di prodotto
param a{risorse,prodotti};
# disponibilità di ciascuna risorsa
param b{risorse};

# VARIABILI
var x {j in prodotti}>=0, <=d[j];

# F.O.
maximize profitto: sum{j in prodotti}c[j]* x[j];

# VINCOLI
subject to diponib {i in risorse}: sum{j in prodotti}a[i,j] * x[j] <= b[i];
```

Script AMPL

Script Esercizio 1

Creare uno script che permetta di calcolare il profitto al variare delle ore disponibili del settore Falegnameria. Le ore possono variare da un minimo di 2000 ad un massimo di 2100 (ad intervalli di 10 ore). Si vuole ottenere, per ogni variazione, un messaggio che comunichi se il corrispondente profitto é accettabile (maggiore di 5450).

Script AMPL

rip_ris.run

```
reset;
model rip_ris.mod;
data rip_ris.dat;
options solver CPLEX;

for {i in 0..10}{
printf "=====\n";
solve;
printf "=====\n";
display i, profitto;
display x;
if (profitto > 5450) then printf "PROFITTO OK!\n";
                        else printf "PROFITTO NON OK!\n";
printf "=====\n";
let b["Falegnameria"]:=b["Falegnameria"]+10;
}
```

Parametri Randomici

Alcuni parametri possono essere generati in maniera randomica:

Dichiarazione(file.mod):

```
param m;
```

```
param v;
```

```
param p:=Normal(m,v);
```

AMPL permette l'utilizzo di diverse funzioni di distribuzione:

Tabella : Funzioni di Distribuzione

Beta(a,b)	distribuzione beta con parametri a e b
Cauchy()	distribuzione di Cauchy
Exponential()	distribuzione esponenziale
Gamma(a)	distribuzione Gamma con parametro a
Normal(m,v)	distribuzione normale con media m e varianza v
Poisson(a)	distribuzione di Poisson con parametro a
Uniform(a,b)	distribuzione uniforme sull'intervallo [a; b]

Possibile scegliere seme per generatore di numeri pseudo-casuali:

```
options randseed; # per mostra il seme
```

```
options randseed value;# cambia il seme
```

Esempio: Problema Investimenti

Esercizio 2

Sono dati $n = 200$ investimenti ciascuno di costo c_i e ritorno r_i . Si chiede di selezionare un sottoinsieme di investimenti in modo tale che il loro prezzo totale non ecceda $B=600$ (budget) e il ritorno sia massimizzato. I costi sono generati con distribuzione uniforme in $[5,20)$ e i ritorni con distribuzione uniforme in $[3,10)$.

Modello Investimenti

- ▶ Modello Matematico Problema Investimenti

$$\begin{aligned} \max \quad & \sum_{i=1}^n r_i x_i \\ & \sum_{i=1}^n c_i x_i \leq B \\ & x_i \in \{0, 1\} \quad i = 1, \dots, n \end{aligned} \tag{1}$$

Modello AMPL

investimenti.mod

```
param N;  
param r{1..N}:= Uniform(5,20); # ritorni  
param c{1..N}:= Uniform(3,10); # costi  
param B; # Budget  
var x{1..N}binary;  
maximize ritorno_tot: sum{i in 1..N}r[i] * x[i];  
subject to budget : sum{i in 1..N}c[i] * x[i] <= B;
```